
In many motion applications the dynamics are such that it is advantageous to be able to 
use different sets of tuning parameters for different parts of a machine cycle and to 
automatically change these parameters on-the-fly.  For example, during one part of a 
cycle, a vertical load is being lifted fighting gravity, while in another part of a cycle that 
same load is being lowered with the aid of gravity.  Another common example is with 
hydraulic cylinders where the system gain is greater during extension than retraction due 
to the area taken up by the cylinder rod. 
 
CTC’s QuickMotion language can easily accommodate these types of applications by 
changing tuning parameters on-the-fly.  Tuning parameter sets can be switched by using 
a foreground MSB (motion sequence block) to change parameters as needed.  Other 
foreground (and background) MSBs trigger this “listening” MSB to change parameters 
before entering a different part of the machine cycle.  This technique is described below 
and when combined with QuickMotion’s deadband control (pdead parameter) the user 
can easily optimize a machine’s performance cycle. 
 
Torque-mode systems 
 
Before coding up the tuning parameter change MSB, 
the user should tune the system by use of the tuning 
wizard for each part of the cycle.  The user should 
mark down the resulting parameters to be entered as 
part of the parameter change MSB.  Once each part 
of the cycle is tuned, the parameters can be 
assembled into the “listening” parameter change 
MSB that will automatically update the tuning 
parameters at the appropriate points of the machine 
cycle. 
 
 
Velocity-mode systems 
 
Velocity-mode systems use only 2 parameters (‘ppg’ and ‘pff’) and do not use the tuning 
wizard.  The method for parameter change is the same, although the number of 
parameters is significantly less and therefore even easier to configure. 

An example of this parameter change MSB for torque-mode: 
 
[top] 
// clear loadparams 
loadparams = 0; 
 
[wlp] 
// loop until loadparams is non-zero 
if loadparams == 0 goto wlp; 
// find out which set to load, and load it 
if loadparams == 1 goto param1; 
if loadparams == 2 goto param2; 
if loadparams == 3 goto param3; 
// loop if an invalid parameter set # was specified 
goto wlp; 
 
// parameter set 1 
[param1] 
kv = 0.015; 
ki = 0.178; 
kvf = 0.9; 
kd = 0.001; 
ppg = 6.5; 
// clear loadparams by heading to [top] 
goto top; 
 
// parameter set 2 
[param2] 
kv = 0.019; 
ki = 0.183; 
kvf = 0.95; 
kd = 0.002; 
ppg = 7.5; 
// clear loadparams by heading to [top] 
goto top; 
 
// parameter set 3 
[param3] 
kv = 0.031; 
ki = 0.128; 
kvf = 0.74; 
kd = 0.001; 
ppg = 3.25; 
// clear loadparams by heading to [top] 
goto top; 
 
// additional parameter sets are added here... 

An example of this parameter change MSB for velocity-mode: 
 
[top] 
// clear loadparams 
loadparams = 0; 
 
[wlp] 
// loop until loadparams is non-zero 
if loadparams == 0 goto wlp; 
// find out which set to load, and load it 
if loadparams == 1 goto param1; 
if loadparams == 2 goto param2; 
if loadparams == 3 goto param3; 
// loop if an invalid parameter set # was specified 
goto wlp; 
 
// parameter set 1 
[param1] 
pff = 0.000; 
ppg = 6.5; 
// clear loadparams by heading to [top] 
goto top; 
 
// parameter set 2 
[param2] 
pff = 0.010; 
ppg = 7.5; 
// clear loadparams by heading to [top] 
goto top; 
 
// parameter set 3 
[param3] 
pff = 0.000; 
ppg = 3.25; 
// clear loadparams by heading to [top] 
goto top; 
 
// additional parameter sets are added here... 

An example on how to use this parameter change MSB: 
 
[top] 
 
// load set #1 
loadparams = 1; 
wait until loadparams == 0; 
 
// do + move 
move in 2.5 to 11.25; 
wait for in position; 
delay 1000; 
 
// load set #2 
loadparams = 2; 
wait until loadparams == 0; 
 
// do - move 
move in 2.5 to 0; 
wait for in position; 
delay 1000; 
 
goto top; 
 


